
Övning 2 – COBOL single SQL  
Uppgift 1 – Include  

000100 Identification Division. 
000200 Program-Id. OVN21. 
000300 Data Division. 
000400 Working-Storage Section. 
000500*--------------------------------------- 
000600* Lägg in INCLUDE av SQL-areor 
000700*--------------------------------------- 
000800     Exec SQL 
000900        Include SQLCA 
001000     End-Exec. 
001100 
001200 Procedure Division. 
001300     Display '-->> Programmet startar...' 
001400     Display '-->> Programmet avslutar...' 
001500 
001600     GoBack 
001700     . 
001800*--------------------------------------- 
001900 End Program OVN21. 
002000*--------------------------------------- 

Uppgift 2 –SQL Select 

000100 Identification Division. 
000200 Program-Id. OVN22. 
000300 Data Division. 
000400 Working-Storage Section. 
000500*--------------------------------------- 
000600* Lägg in INCLUDE av SQL-areor 
000700* och beskrivningar. 
000800*--------------------------------------- 
000900     Exec SQL 
001000        Include SQLCA 
001100     End-Exec. 
001200 
001300 01 SQL-Data. 
001400    05 WS-Datum      Pic X(10). 
001500    05 WS-Tid        Pic X(10). 
001600    05 WS-SQLCODE    Pic +(09). 
001700 
001800 
001900 Procedure Division. 
002000     Display '-->> Programmet startar...' 
002100     Perform Hamta-Datum-Tid 
002200     Perform Visa-Datum-Tid 
002300     Display '-->> Programmet avslutar...' 
002400 
002500     GoBack 
002600     . 
002700*--------------------------------------- 
002800 Hamta-Datum-Tid. 
002900*--------------------------------------- 

003000     Exec SQL 
003100        Select 
003200           CURRENT DATE, 
003300           CURRENT TIME 
003400        Into 
003500           :WS-Datum, 
003600           :WS-Tid 
003700        From 
003800           SYSIBM.SYSDUMMY1 
003900     End-Exec 
004000     . 
004100*--------------------------------------- 
004200 Visa-Datum-Tid. 
004300*--------------------------------------- 
004400     Display '-->> Datum : ' WS-Datum 
004500     Display '-->> Tid   : ' WS-Tid 
004600     . 
004700*--------------------------------------- 
004800 End Program OVN22. 
004900*--------------------------------------- 

Övning 3 – COBOL CURSOR  
Uppgift 1 – Include  

000100 Identification Division. 
000200 Program-Id. OVN31. 
000300 Data Division. 
000400 Working-Storage Section. 
000500*--------------------------------------- 
000600* Lägg in INCLUDE av SQL-areor 
000700* och beskrivningar. 
000800*--------------------------------------- 
000900     Exec SQL 
001000        Include SQLCA 
001100     End-Exec. 
001200 
001300     Exec SQL 
001400        Include DRYCK 
001500     End-Exec. 
001600 
001700     Exec SQL 
001800        Include PRIS 
001900     End-Exec. 
002000 
002100 01 SQL-Data. 
002200    05 WS-Datum      Pic X(10). 
002300    05 WS-Tid        Pic X(10). 
002400    05 WS-SQLCODE    Pic +(09). 
002500 
002600 
002700 Procedure Division. 
002800     Display '-->> Programmet startar...' 
002900     Perform Hamta-Datum-Tid 
003000     Perform Visa-Datum-Tid 
003100     Display '-->> Programmet avslutar...' 
003200 
003300     GoBack 
003400     . 
003500*--------------------------------------- 
003600 Hamta-Datum-Tid. 
003700*--------------------------------------- 
003800     Exec SQL 
003900        Select 
004000           CURRENT DATE, 
004100           CURRENT TIME 
004200        Into 
004300           :WS-Datum, 
004400           :WS-Tid 
004500        From 
004600           SYSIBM.SYSDUMMY1 
004700     End-Exec 
004800     . 
004900*--------------------------------------- 
005000 Visa-Datum-Tid. 
005100*--------------------------------------- 
005200     Display '-->> Datum : ' WS-Datum 
005300     Display '-->> Tid   : ' WS-Tid 
005400     . 

005500*--------------------------------------- 
005600 End Program OVN31. 
005700*--------------------------------------- 
 

Uppgift 2 – Parameterdatum  

000100 Identification Division. 
000200 Program-Id. OVN32. 
000300 Data Division. 
000400 Working-Storage Section. 
000500*--------------------------------------- 
000600* Lägg in INCLUDE av SQL-areor 
000700* och beskrivningar. 
000800*--------------------------------------- 
000900     Exec SQL 
001000        Include SQLCA 
001100     End-Exec. 
001200 
001300     Exec SQL 
001400        Include DRYCK 
001500     End-Exec. 
001600 
001700     Exec SQL 
001800        Include PRIS 
001900     End-Exec. 
002000 
002100 01 SQL-Data. 
002200    05 WS-Datum      Pic X(10). 
002300    05 WS-Tid        Pic X(10). 
002400    05 WS-SQLCODE    Pic +(09). 
002500 
002600 01 SysinData. 
002700    05 WS-Parameter-Datum Pic X(10). 
002800 
002900 
003000 Procedure Division. 
003100     Display '-->> Programmet startar...' 
003200     Perform Hamta-Datum-Tid 
003300     Perform Hamta-Parameter-Datum 
003400     Perform Visa-Datum-Tid 
003500     Display '-->> Programmet avslutar...' 
003600 
003700     GoBack 
003800     . 
003900*--------------------------------------- 
004000 Hamta-Datum-Tid. 
004100*--------------------------------------- 
004200     Exec SQL 
004300        Select 
004400           CURRENT DATE, 
004500           CURRENT TIME 
004600        Into 
004700           :WS-Datum, 
004800           :WS-Tid 
004900        From 
005000           SYSIBM.SYSDUMMY1 
005100     End-Exec 

005200     . 
005300*--------------------------------------- 
005400 Hamta-Parameter-Datum. 
005500*--------------------------------------- 
005600     Accept WS-Parameter-Datum From SYSIN 
005700     . 
005800*--------------------------------------- 
005900 Visa-Datum-Tid. 
006000*--------------------------------------- 
006100     Display '-->> Datum  : ' WS-Datum 
006200     Display '-->> Tid    : ' WS-Tid 
006300     Display '-->> Pdatum : ' WS-Parameter-Datum 
006400     . 
006500*--------------------------------------- 
006600 End Program OVN32. 
006700*--------------------------------------- 

 

Uppgift 3 – Prislista  

000100 Identification Division. 
000200 Program-Id. OVN33. 
000300 Data Division. 
000400 Working-Storage Section. 
000500*--------------------------------------- 
000600* Lägg in INCLUDE av SQL-areor 
000700* och beskrivningar. 
000800*--------------------------------------- 
000900     Exec SQL 
001000        Include SQLCA 
001100     End-Exec. 
001200 
001300     Exec SQL 
001400        Include DRYCK 
001500     End-Exec. 
001600 
001700     Exec SQL 
001800        Include PRIS 
001900     End-Exec. 
002000 
002100     Exec SQL 
002200        DECLARE PRISER CURSOR FOR 
002300           SELECT 
002400              D.DRYCKNR, 
002500              D.DRYCKNAMN, 
002600              P.VOLYM, 
002700              P.PRIS, 
002800              P.PRDATUM 
002900           FROM 
003000              UTB00.DRYCK D JOIN UTB00.PRIS P 
003100              ON D.DRYCKNR = P.DRYCKNR 
003200           WHERE 
003300              (P.DRYCKNR, P.VOLYM, P.PRDATUM) IN 
003400                 (SELECT 
003500                    P2.DRYCKNR, 
003600                    P2.VOLYM, 

003700                    MAX(P2.PRDATUM) 
003800                  FROM 
003900                     UTB00.PRIS P2 
004000                  WHERE 
004100                     P2.PRDATUM <= :WS-Sysin-Datum 
004200                  GROUP BY 
004300                     P2.DRYCKNR 
004400                    ,P2.VOLYM) 
004500                  ORDER BY D.DRYCKNR, P.VOLYM 
004600     End-Exec. 
004700 
004800 01 SQL-Data. 
004900    05 WS-Datum      Pic X(10). 
005000    05 WS-Tid        Pic X(10). 
005100    05 WS-SQLCODE    Pic +(09). 
005200 
005300 01 SysinData. 
005400    05 WS-Sysin-Datum Pic X(10). 
005500 
005600 01 Display-Rad. 
005700    05 D-Drycknr     Pic 9(05). 
005800    05               Pic X(02) Value Space. 
005900    05 D-Drycknamn   Pic X(25). 
006000    05               Pic X(02) Value Space. 
006100    05 D-Volym       Pic 9(05). 
006200    05               Pic X(02) Value Space. 
006300    05 D-Pris        Pic 9(05). 
006400    05               Pic X(02) Value Space. 
006500    05 D-Prdatum     Pic X(10). 
006600 
006700 01 Vilkoren         Pic X(08) Value Space. 
006800    88 Klar                    Value 'Klar'. 
006900 
007000 
007100 Procedure Division. 
007200     Display '-->> Programmet startar...' 
007300     Perform Hamta-Datum-Tid 
007400     Perform Hamta-Sysin-Datum 
007500     Perform Visa-Datum-Tid 
007600     Perform SQL-Oppna-Cursor 
007700     Perform SQL-Fetch-Cursor Until Klar 
007800     Display '-->> Programmet avslutar...' 
007900 
008000     GoBack 
008100     . 
008200*--------------------------------------- 
008300 Hamta-Datum-Tid. 
008400*--------------------------------------- 
008500     Exec SQL 
008600        Select 
008700           CURRENT DATE, 
008800           CURRENT TIME 
008900        Into 
009000           :WS-Datum, 
009100           :WS-Tid 
009200        From 
009300           SYSIBM.SYSDUMMY1 
009400     End-Exec 
009500     . 
009600*--------------------------------------- 
009700 Hamta-Sysin-Datum. 

009800*--------------------------------------- 
009900     Accept WS-Sysin-Datum From SYSIN 
010000     . 
010100*--------------------------------------- 
010200 SQL-Oppna-Cursor. 
010300*--------------------------------------- 
010400     Exec SQL 
010500         OPEN PRISER 
010600     End-Exec 
010700     . 
010800*--------------------------------------- 
010900 SQL-Fetch-Cursor. 
011000*--------------------------------------- 
011100     Exec SQL 
011200         FETCH PRISER 
011300           INTO 
011400            :DCLDRYCK.DRYCKNR, 
011500            :DCLDRYCK.DRYCKNAMN, 
011600            :DCLPRIS.VOLYM, 
011700            :DCLPRIS.PRIS, 
011800            :DCLPRIS.PRDATUM 
011900     End-Exec 
012000 
012100     Evaluate SQLCODE 
012200        When 0 
012300           Perform Display-Rader 
012400        When 100 
012500           Set Klar to True 
012600        When Other 
012700           Perform SQL-Fel 
012800     End-Evaluate 
012900     . 
013000*--------------------------------------- 
013100 Stang-Cursor. 
013200*--------------------------------------- 
013300     Exec SQL 
013400         CLOSE PRISER 
013500     End-Exec 
013600     . 
013700*--------------------------------------- 
013800 Display-Rader. 
013900*--------------------------------------- 
014000     Move Drycknr   in DclDryck to D-Drycknr 
014100     Move Drycknamn in DclDryck to D-Drycknamn 
014200     Move Volym     in DclPris  to D-Volym 
014300     Move Pris      in DclPris  to D-Pris 
014400     Move Prdatum   in DclPris  to D-Prdatum 
014500     Display Display-Rad 
014600     . 
014700*--------------------------------------- 
014800 Visa-Datum-Tid. 
014900*--------------------------------------- 
015000     Display '-->> Datum  : ' WS-Datum 
015100     Display '-->> Tid    : ' WS-Tid 
015200     Display '-->> Pdatum : ' WS-Sysin-Datum 
015300     Display Space 
015400     . 
015500*--------------------------------------- 
015600 SQL-Fel. 
015700*--------------------------------------- 
015800     Move SQLCODE to WS-SQLCODE 

015900     Display '*** SQL-Fel ***' 
016000     Display 'SQLCODE  : ' WS-SQLCODE 
016100     Display 'SQLSTATE : ' SQLSTATE 
016200     Display 'SQLMedd  : ' SQLERRMC(1:SQLERRML) 
016300     Move 99 to Return-Code 
016400     Set Klar to True 
016500     . 
016600*--------------------------------------- 
016700 End Program OVN33. 
016800*--------------------------------------- 
 

Uppgift 4 – Prislista till fil 

000100 Identification Division. 
000200 Program-Id. OVN34. 
000300 Environment Division. 
000400 Input-Output Section. 
000500 File-Control. 
000600     Select Lista Assign to LISTDD 
000700         Organization is Sequential 
000800         Access is Sequential 
000900         . 
001000 
001100 Data Division. 
001200 File Section. 
001300 FD Lista Recording Mode F. 
001400 01 Listrad   Pic X(120). 
001500 
001600 Working-Storage Section. 
001700     Exec SQL 
001800        Include SQLCA 
001900     End-Exec. 
002000 
002100     Exec SQL 
002200        Include DRYCK 
002300     End-Exec. 
002400 
002500     Exec SQL 
002600        Include PRIS 
002700     End-Exec. 
002800 
002900     Exec SQL 
003000        DECLARE PRISER CURSOR FOR 
003100           SELECT 
003200              D.DRYCKNR, 
003300              D.DRYCKNAMN, 
003400              P.VOLYM, 
003500              P.PRIS, 
003600              P.PRDATUM 
003700           FROM 
003800              UTB00.DRYCK D JOIN UTB00.PRIS P 
003900              ON D.DRYCKNR = P.DRYCKNR 
004000           WHERE 
004100              (P.DRYCKNR, P.VOLYM, P.PRDATUM) IN 
004200                 (SELECT 
004300                    P2.DRYCKNR, 
004400                    P2.VOLYM, 

004500                    MAX(P2.PRDATUM) 
004600                  FROM 
004700                     UTB00.PRIS P2 
004800                  WHERE 
004900                     P2.PRDATUM <= :WS-Sysin-Datum 
005000                  GROUP BY 
005100                     P2.DRYCKNR 
005200                    ,P2.VOLYM) 
005300                  ORDER BY D.DRYCKNR, P.VOLYM 
005400     End-Exec. 
005500 
005600 01 WS-Data. 
005700    05 WS-Sidnr      Pic 9(02) Value 0. 
005800    05 WS-Radnr      Pic 9(02) Value 0. 
005900       88 Radnr-Start          Value 0. 
006000       88 Radnr-Max            Value 15. 
006100 
006200 01 SQL-Data. 
006300    05 WS-Datum      Pic X(10). 
006400    05 WS-Tid        Pic X(10). 
006500    05 WS-SQLCODE    Pic +(09). 
006600 
006700 01 SysinData. 
006800    05 WS-Sysin-Datum Pic X(10). 
006900 
007000 01 Prislista-Dryckrad. 
007100    05 D-Drycknr     Pic 9(05). 
007200    05               Pic X(03) Value Space. 
007300    05 D-Drycknamn   Pic X(25). 
007400    05               Pic X(02) Value Space. 
007500    05 D-Volym       Pic 9(05). 
007600    05               Pic X(02) Value Space. 
007700    05 D-Pris        Pic 9(05). 
007800    05               Pic X(02) Value Space. 
007900    05 D-Prdatum     Pic X(10). 
008000 
008100 01 Prislista-Rubrik. 
008200    05          Pic X(08) Value 'Drycknr'. 
008300    05          Pic X(27) Value 'Drycknamn'. 
008400    05          Pic X(07) Value 'Volym'. 
008500    05          Pic X(07) Value 'Pris'. 
008600    05          Pic X(12) Value 'Prisdatum'. 
008700    05          Pic X(05) Value 'Sida:'. 
008800    05 R-Sidnr  Pic Z(02). 
008900 
009000 01 Prislista-Slutrad. 
009100    05  Pic X(30) Value 'S L U T på Prislista'. 
009200 
009300 01 Prislista-Skiljerad. 
009400    05  Pic X(120) Value All '-'. 
009500 
009600 01 Vilkoren         Pic X(08) Value Space. 
009700    88 Klar                    Value 'Klar'. 
009800 
009900 
010000 Procedure Division. 
010100     Display '-->> Programmet startar...' 
010200     Perform Initiera 
010300     Perform SQL-Oppna-Cursor 
010400     Perform SQL-Fetch-Cursor Until Klar 
010500     Perform Avsluta 

010600     Display '-->> Programmet avslutar...' 
010700 
010800     GoBack 
010900     . 
011000*--------------------------------------- 
011100 Initiera. 
011200*--------------------------------------- 
011300     Open Output Lista 
011400     Accept WS-Sysin-Datum From SYSIN 
011500     . 
011600*--------------------------------------- 
011700 Avsluta. 
011800*--------------------------------------- 
011900     Perform Skriv-Prislista-Slutrad 
012000     Close Lista 
012100     . 
012200*--------------------------------------- 
012300 SQL-Oppna-Cursor. 
012400*--------------------------------------- 
012500     Exec SQL 
012600         OPEN PRISER 
012700     End-Exec 
012800     . 
012900*--------------------------------------- 
013000 SQL-Fetch-Cursor. 
013100*--------------------------------------- 
013200     Exec SQL 
013300         FETCH PRISER 
013400           INTO 
013500            :DCLDRYCK.DRYCKNR, 
013600            :DCLDRYCK.DRYCKNAMN, 
013700            :DCLPRIS.VOLYM, 
013800            :DCLPRIS.PRIS, 
013900            :DCLPRIS.PRDATUM 
014000     End-Exec 
014100 
014200     Evaluate SQLCODE 
014300        When 0 
014400           Perform Skriv-Prislista-Rad 
014500        When 100 
014600           Set Klar to True 
014700        When Other 
014800           Perform SQL-Fel 
014900     End-Evaluate 
015000     . 
015100*--------------------------------------- 
015200 Stang-Cursor. 
015300*--------------------------------------- 
015400     Exec SQL 
015500         CLOSE PRISER 
015600     End-Exec 
015700     . 
015800*--------------------------------------- 
015900 Skriv-Prislista-Rad. 
016000*--------------------------------------- 
016100     Evaluate True 
016200        When WS-Sidnr = 0 
016300           Perform Skriv-Prislista-Rubrik 
016400        When Radnr-Max 
016500           Perform Skriv-Prislista-Rubrik 
016600           Set Radnr-Start to True 

016700     End-Evaluate 
016800     Perform Skriv-Prislista-Dryckrad 
016900     . 
017000*--------------------------------------- 
017100 Skriv-Prislista-Rubrik. 
017200*--------------------------------------- 
017300     Add 1 to WS-Sidnr 
017400     Move WS-Sidnr to R-Sidnr 
017500     Write Listrad 
017600           From Prislista-Rubrik 
017700     Write Listrad 
017800           From Prislista-Skiljerad 
017900     . 
018000*--------------------------------------- 
018100 Skriv-Prislista-Dryckrad. 
018200*--------------------------------------- 
018300     Add 1 to WS-Radnr 
018400     Move Drycknr   in DclDryck to D-Drycknr 
018500     Move Drycknamn in DclDryck to D-Drycknamn 
018600     Move Volym     in DclPris  to D-Volym 
018700     Move Pris      in DclPris  to D-Pris 
018800     Move Prdatum   in DclPris  to D-Prdatum 
018900     Write Listrad From Prislista-Dryckrad 
019000     . 
019100*--------------------------------------- 
019200 Skriv-Prislista-Slutrad. 
019300*--------------------------------------- 
019400     Write Listrad 
019500           From Prislista-Skiljerad 
019600     Write Listrad 
019700           From Prislista-Slutrad 
019800     . 
019900*--------------------------------------- 
020000 SQL-Fel. 
020100*--------------------------------------- 
020200     Move SQLCODE to WS-SQLCODE 
020300     Display '*** SQL-Fel ***' 
020400     Display 'SQLCODE  : ' WS-SQLCODE 
020500     Display 'SQLSTATE : ' SQLSTATE 
020600     Display 'SQLMedd  : ' SQLERRMC(1:SQLERRML) 
020700     Move 99 to Return-Code 
020800     Set Klar to True 
020900     . 
021000*--------------------------------------- 
021100 End Program OVN34. 
021200*--------------------------------------- 
 

Uppgift 5 – Prislista till fil,med brytning 

000100 Identification Division. 
000200 Program-Id. OVN35. 
000300 Environment Division. 
000400 Input-Output Section. 
000500 File-Control. 
000600     Select Lista Assign to LISTDD 
000700         Organization is Sequential 
000800         Access is Sequential 

000900         . 
001000 
001100 Data Division. 
001200 File Section. 
001300 FD Lista Recording Mode F. 
001400 01 Listrad   Pic X(120). 
001500 
001600 Working-Storage Section. 
001700     Exec SQL 
001800        Include SQLCA 
001900     End-Exec. 
002000 
002100     Exec SQL 
002200        Include DRYCK 
002300     End-Exec. 
002400 
002500     Exec SQL 
002600        Include PRIS 
002700     End-Exec. 
002800 
002900     Exec SQL 
003000        DECLARE PRISER CURSOR FOR 
003100           SELECT 
003200              D.DRYCKNR, 
003300              D.DRYCKNAMN, 
003400              D.DSKOD, 
003500              P.VOLYM, 
003600              P.PRIS, 
003700              P.PRDATUM 
003800           FROM 
003900              UTB00.DRYCK D JOIN UTB00.PRIS P 
004000              ON D.DRYCKNR = P.DRYCKNR 
004100           WHERE 
004200              (P.DRYCKNR, P.VOLYM, P.PRDATUM) IN 
004300                 (SELECT 
004400                    P2.DRYCKNR, 
004500                    P2.VOLYM, 
004600                    MAX(P2.PRDATUM) 
004700                  FROM 
004800                     UTB00.PRIS P2 
004900                  WHERE 
005000                     P2.PRDATUM <= :WS-Sysin-Datum 
005100                  GROUP BY 
005200                     P2.DRYCKNR 
005300                    ,P2.VOLYM) 
005400                  ORDER BY D.DSKOD, D.DRYCKNR, 
P.VOLYM 
005500     End-Exec. 
005600 
005700 01 WS-Data. 
005800    05 WS-Sidnr      Pic 9(02) Value 0. 
005900    05 WS-Radnr      Pic 9(02) Value 0. 
006000       88 Radnr-Start          Value 0. 
006100       88 Radnr-Max            Value 55. 
006200    05 WS-Old-Dskod  Pic X(02) Value Space. 
006300 
006400 01 SQL-Data. 
006500    05 WS-Datum      Pic X(10). 
006600    05 WS-Tid        Pic X(10). 
006700    05 WS-SQLCODE    Pic +(09). 
006800 

006900 01 SysinData. 
007000    05 WS-Sysin-Datum Pic X(10). 
007100 
007200 01 Prislista-Dryckrad. 
007300    05 D-Drycknr     Pic 9(05). 
007400    05               Pic X(03) Value Space. 
007500    05 D-Drycknamn   Pic X(25). 
007600    05               Pic X(02) Value Space. 
007700    05 D-Volym       Pic 9(05). 
007800    05               Pic X(02) Value Space. 
007900    05 D-Pris        Pic 9(05). 
008000    05               Pic X(02) Value Space. 
008100    05 D-Prdatum     Pic X(10). 
008200 
008300 01 Prislista-Rubrik. 
008400    05          Pic X(08) Value 'Drycknr'. 
008500    05          Pic X(27) Value 'Drycknamn'. 
008600    05          Pic X(07) Value 'Volym'. 
008700    05          Pic X(07) Value 'Pris'. 
008800    05          Pic X(12) Value 'Prisdatum'. 
008900    05          Pic X(05) Value 'Sida:'. 
009000    05 R-Sidnr  Pic Z(02). 
009100 
009200 01 Prislista-Dsnamn. 
009300    05          Pic X(06) Value '-->> ('. 
009400    05 Dskod    Pic X(02). 
009500    05          Pic X(04) Value ')'. 
009600    05 Dsnamn   Pic X(40). 
009700 
009800 01 Prislista-Slutrad. 
009900    05  Pic X(30) Value 'S L U T på Prislista'. 
010000 
010100 01 Prislista-Skiljerad. 
010200    05  Pic X(120) Value All '-'. 
010300 
010400 01 Prislista-Tomrad. 
010500    05  Pic X(120) Value Space. 
010600 
010700 01 Vilkoren         Pic X(08) Value Space. 
010800    88 Klar                    Value 'Klar'. 
010900 
011100 Procedure Division. 
011200     Display '-->> Programmet startar...' 
011300     Perform Initiera 
011400     Perform SQL-Oppna-Cursor 
011500     Perform SQL-Fetch-Cursor Until Klar 
011600     Perform Avsluta 
011700     Display '-->> Programmet avslutar...' 
011800 
011900     GoBack 
012000     . 
012100*--------------------------------------- 
012200 Initiera. 
012300*--------------------------------------- 
012400     Open Output Lista 
012500     Accept WS-Sysin-Datum From SYSIN 
012600     . 
012700*--------------------------------------- 
012800 Avsluta. 
012900*--------------------------------------- 
013000     Perform Skriv-Prislista-Slutrad 

013100     Close Lista 
013200     . 
013300*--------------------------------------- 
013400 SQL-Oppna-Cursor. 
013500*--------------------------------------- 
013600     Exec SQL 
013700         OPEN PRISER 
013800     End-Exec 
013900     . 
014000*--------------------------------------- 
014100 SQL-Fetch-Cursor. 
014200*--------------------------------------- 
014300     Exec SQL 
014400         FETCH PRISER 
014500           INTO 
014600            :DCLDRYCK.DRYCKNR, 
014700            :DCLDRYCK.DRYCKNAMN, 
014800            :DCLDRYCK.DSKOD, 
014900            :DCLPRIS.VOLYM, 
015000            :DCLPRIS.PRIS, 
015100            :DCLPRIS.PRDATUM 
015200     End-Exec 
015300 
015400     Evaluate SQLCODE 
015500        When 0 
015600           Perform Skriv-Prislista-Rad 
015700        When 100 
015800           Set Klar to True 
015900        When Other 
016000           Perform SQL-Fel 
016100     End-Evaluate 
016200     . 
016300*--------------------------------------- 
016400 Stang-Cursor. 
016500*--------------------------------------- 
016600     Exec SQL 
016700         CLOSE PRISER 
016800     End-Exec 
016900     . 
017000*--------------------------------------- 
017100 Skriv-Prislista-Rad. 
017200*--------------------------------------- 
017300     Evaluate True 
017400        When WS-Sidnr = 0 
017500           Perform Skriv-Prislista-Rubrik 
017600           Perform Skriv-Prislista-Dsnamn 
017700        When WS-Old-Dskod Not = Dskod in DclDryck 
017800           Perform Skriv-Prislista-Rubrik 
017900           Perform Skriv-Prislista-Dsnamn 
018000        When Radnr-Max 
018100           Perform Skriv-Prislista-Rubrik 
018200           Set Radnr-Start to True 
018300     End-Evaluate 
018400     Perform Skriv-Prislista-Dryckrad 
018500     . 
018600*--------------------------------------- 
018700 Skriv-Prislista-Rubrik. 
018800*--------------------------------------- 
018900     Add 1 to WS-Sidnr 
019000     Move WS-Sidnr to R-Sidnr 
019100     Write Listrad 

019200           From Prislista-Tomrad 
019300     Write Listrad 
019400           From Prislista-Rubrik 
019500     Write Listrad 
019600           From Prislista-Skiljerad 
019700     . 
019800*--------------------------------------- 
019900 Skriv-Prislista-Dryckrad. 
020000*--------------------------------------- 
020100     Add 1 to WS-Radnr 
020200     Move Drycknr   in DclDryck to D-Drycknr 
020300     Move Drycknamn in DclDryck to D-Drycknamn 
020400     Move Volym     in DclPris  to D-Volym 
020500     Move Pris      in DclPris  to D-Pris 
020600     Move Prdatum   in DclPris  to D-Prdatum 
020700     Write Listrad From Prislista-Dryckrad 
020800     . 
020900*--------------------------------------- 
021000 Skriv-Prislista-Dsnamn. 
021100*--------------------------------------- 
021200     Exec SQL 
021300          Select 
021400             Dsnamn 
021500          Into 
021600             :Prislista-Dsnamn.Dsnamn 
021700          From 
021800             UTB00.DRSLAG 
021900          Where 
022000             Dskod = :DclDryck.Dskod 
022100     End-Exec 
022200     If SQLCODE = +100 
022300        Move '*** I/U ***' to Dsnamn  
022310              in Prislista-Dsnamn 
022400     End-If 
022500     Move Dskod in DclDryck to WS-Old-Dskod 
022600                   Dskod in Prislista-Dsnamn 
022700     Write Listrad from Prislista-Tomrad 
022800     Write Listrad from Prislista-Dsnamn 
022900     . 
023000*--------------------------------------- 
023100 Skriv-Prislista-Slutrad. 
023200*--------------------------------------- 
023300     Write Listrad 
023400           From Prislista-Skiljerad 
023500     Write Listrad 
023600           From Prislista-Slutrad 
023700     . 
023800*--------------------------------------- 
023900 SQL-Fel. 
024000*--------------------------------------- 
024100     Move SQLCODE to WS-SQLCODE 
024200     Display '*** SQL-Fel ***' 
024300     Display 'SQLCODE  : ' WS-SQLCODE 
024400     Display 'SQLSTATE : ' SQLSTATE 
024500     Display 'SQLMedd  : ' SQLERRMC(1:SQLERRML) 
024600     Move 99 to Return-Code 
024700     Set Klar to True 
024800     . 
024900*--------------------------------------- 
025000 End Program OVN35. 

025100*--------------------------------------- 

Övning 4 – COBOL INSERT  
Uppgift 1 – Kontroll av indata  

000100 Identification Division. 
000200 Program-Id. OVN41. 
000300 Environment Division. 
000400 Input-Output Section. 
000500 File-Control. 
000600     Select DATAIN Assign to DATADD 
000700         Organization is Sequential 
000800         Access is Sequential 
000900         . 
001000 Data Division. 
001100 
001200 File Section. 
001300 FD Datain Recording Mode F. 
001400 01 Datain-Post Pic X(80). 
001500 
001600 Working-Storage Section. 
001700     Exec SQL 
001800        Include SQLCA 
001900     End-Exec 
002000     . 
002100     Exec SQL 
002200        Include DRYCK 
002300     End-Exec 
002400     . 
002500     Exec SQL 
002600        Include PRIS 
002700     End-Exec 
002800     . 
002900 01 WS-Datain-Post. 
003000    05 WS-In-Drycknr      Pic 9(05). 
003100    05                    Pic X(06). 
003200    05 WS-In-Volym        Pic 9(04). 
003300    05                    Pic X(05). 
003400    05 WS-In-Pris         Pic 9(04). 
003500    05                    Pic X(05). 
003600    05 WS-In-Prdatum      Pic X(10). 
003700 
003800 01 Villkor        Pic 9(01) Value 0. 
003900    88 Klar                  Value 1. 
004000 
004100 01 SQL-Data. 
004200    05 WS-SQLCODE  Pic +(09). 
004300 
004400 
004500 Procedure Division. 
004600     Display '-->> Programmet startar...' 
004700     Perform Oppna-Datain 
004800     Perform Laes-Datain Until Klar 
004900     Perform Stang-Datain 
005000     Display '-->> Programmet avslutar...' 
005100 
005200     GoBack 
005300     . 
005400*--------------------------------------- 
005500 Oppna-Datain. 

005600*--------------------------------------- 
005700     Open Input DATAIN 
005800     Display '-->> Datain öppnad...' 
005900     . 
006000*--------------------------------------- 
006100 Laes-Datain. 
006200*--------------------------------------- 
006300     Read DATAIN Into WS-Datain-Post 
006400        At End 
006500           Set Klar to True 
006600        Not At End 
006700           Perform Behandla-Datain-Post 
006800     End-Read 
006900     . 
007000*--------------------------------------- 
007100 Behandla-Datain-Post. 
007200*--------------------------------------- 
007300 
007400     Move WS-In-Drycknr to Drycknr in Dcldryck 
007500 
007600     Exec SQL 
007700       Select 
007800          Drycknamn 
007900       Into 
008000          :Drycknamn 
008100       From 
008200          UTB00.DRYCK 
008300       Where 
008400          Drycknr = :Dcldryck.Drycknr 
008500     End-Exec 
008600 
008700     Evaluate SQLCODE 
008800        When 0 
008900           Continue 
009000        When 100 
009100           Display '*** Fel Drycknr: '  
009110                        WS-In-Drycknr 
009200        When Other 
009300           Perform SQL-Fel 
009400     End-Evaluate 
009500     . 
009600*--------------------------------------- 
009700 Stang-Datain. 
009800*--------------------------------------- 
009900     Close Datain 
010000     Display '-->> Datain stängd...' 
010100     . 
010200*--------------------------------------- 
010300 SQL-Fel. 
010400*--------------------------------------- 
010500     Move SQLCODE to WS-SQLCODE 
010600     Display '*** SQL-Fel ***' 
010700     Display 'SQLCODE  : ' WS-SQLCODE 
010800     Display 'SQLSTATE : ' SQLSTATE 
010900     Display 'SQLMedd  : ' SQLERRMC(1:SQLERRML) 
011000     Move 99 to Return-Code 
011100     Set KLAR to True 
011200     . 
011300*--------------------------------------- 
011400 End Program OVN41. 

011500*--------------------------------------- 

 

Uppgift 2 – INSERT av godkända rader  

000100 Identification Division. 
000200 Program-Id. OVN42. 
000300 Environment Division. 
000400 Input-Output Section. 
000500 File-Control. 
000600     Select DATAIN Assign to DATADD 
000700         Organization is Sequential 
000800         Access is Sequential 
000900         . 
001000 Data Division. 
001100 
001200 File Section. 
001300 FD Datain Recording Mode F. 
001400 01 Datain-Post Pic X(80). 
001500 
001600 Working-Storage Section. 
001700     Exec SQL 
001800        Include SQLCA 
001900     End-Exec 
002000     . 
002100     Exec SQL 
002200        Include DRYCK 
002300     End-Exec 
002400     . 
002500     Exec SQL 
002600        Include PRIS 
002700     End-Exec 
002800     . 
002900 01 WS-Datain-Post. 
003000    05 WS-In-Drycknr Pic 9(05). 
003100    05               Pic X(06). 
003200    05 WS-In-Volym   Pic 9(04). 
003300    05               Pic X(05). 
003400    05 WS-In-Pris    Pic 9(04). 
003500    05               Pic X(05). 
003600    05 WS-In-Prdatum Pic X(10). 
003700 
003800 01 Villkor          Pic 9(01) Value 0. 
003900    88 Klar                    Value 1. 
004000 
004100 01 SQL-Data. 
004200    05 WS-SQLCODE     Pic +(09). 
004300    05 WS-Antal-Rader Pic 9(04) Value 0. 
004400 
004500 
004600 Procedure Division. 
004700     Display '-->> Programmet startar...' 
004800     Perform Oppna-Datain 
004900     Perform Laes-Datain Until Klar 
005000     Perform Stang-Datain 
005100     Display '-->> ' WS-Antal-Rader  

005110                   ' rader behandlade' 
005200     Display '-->> Programmet avslutar...' 
005300 
005400     GoBack 
005500     . 
005600*--------------------------------------- 
005700 Oppna-Datain. 
005800*--------------------------------------- 
005900     Open Input DATAIN 
006000     Display '-->> Datain öppnad...' 
006100     . 
006200*--------------------------------------- 
006300 Laes-Datain. 
006400*--------------------------------------- 
006500     Read DATAIN Into WS-Datain-Post 
006600        At End 
006700           Set Klar to True 
006800        Not At End 
006900           Perform Behandla-Datain-Post 
007000     End-Read 
007100     . 
007200*--------------------------------------- 
007300 Behandla-Datain-Post. 
007400*--------------------------------------- 
007500 
007600     Add 1 to WS-Antal-Rader 
007700     Move WS-In-Drycknr to Drycknr in Dcldryck 
007800 
007900     Exec SQL 
008000       Select 
008100          Drycknamn 
008200       Into 
008300          :Drycknamn 
008400       From 
008500          UTB00.DRYCK 
008600       Where 
008700          Drycknr = :Dcldryck.Drycknr 
008800     End-Exec 
008900 
009000     Evaluate SQLCODE 
009100        When 0 
009200           Perform SQL-Insert-Pris 
009300        When 100 
009400           Display '*** Fel Drycknr: '  
009410                    WS-In-Drycknr 
009500        When Other 
009600           Display '*** Annat fel vid Select '  
009610                    WS-In-Drycknr 
009700           Perform SQL-Fel 
009800     End-Evaluate 
009900     . 
010000*--------------------------------------- 
010100 SQL-Insert-Pris. 
010200*--------------------------------------- 
010400     Move Ws-In-Volym to Volym in DclPris 
010500     Move Ws-In-Pris  to Pris  in DclPris 
010600     Move Ws-In-Prdatum to Prdatum in DclPris 
010700     Exec SQL 
010800     Insert 
010900        Into UTB00.Pris 
011000     Values 

011100        (:Dcldryck.Drycknr, 
011200         :Volym, 
011300         :Pris, 
011400          Date(:Prdatum)) 
011500     End-Exec 
011600 
011700     Evaluate SQLCODE 
011800        When 0 
011900*          Display '+++ Drycknr : ' Ws-In-Drycknr 
012000*                  ' nytt prisdatum ' Ws-In-Prdatum 
012100        Continue 
012200        When Other 
012300           Perform SQL-Fel 
012400     End-Evaluate 
012500     . 
012600*--------------------------------------- 
012700 Stang-Datain. 
012800*--------------------------------------- 
012900     Close Datain 
013000     Display '-->> Datain stängd...' 
013100     . 
013200*--------------------------------------- 
013300 SQL-Fel. 
013400*--------------------------------------- 
013500     Move SQLCODE to WS-SQLCODE 
013600     Display '*** SQL-Fel ***' 
013700     Display 'SQLCODE  : ' WS-SQLCODE 
013800     Display 'SQLSTATE : ' SQLSTATE 
013900     Display 'SQLMedd  : ' SQLERRMC(1:SQLERRML) 
014000     Move 99 to Return-Code 
014100     Set KLAR to True 
014200     . 
014300*--------------------------------------- 
014400 End Program OVN42. 
014500*--------------------------------------- 
 

 

Uppgift 3 – INSERT med felhantering  

000100 Identification Division. 
000200 Program-Id. OVN43. 
000300 Environment Division. 
000400 Input-Output Section. 
000500 File-Control. 
000600     Select DATAIN Assign to DATADD 
000700         Organization is Sequential 
000800         Access is Sequential 
000900         . 
001000 Data Division. 
001100 
001200 File Section. 
001300 FD Datain Recording Mode F. 
001400 01 Datain-Post Pic X(80). 
001500 
001600 Working-Storage Section. 

001700     Exec SQL 
001800        Include SQLCA 
001900     End-Exec 
002000     . 
002100     Exec SQL 
002200        Include DRYCK 
002300     End-Exec 
002400     . 
002500     Exec SQL 
002600        Include PRIS 
002700     End-Exec 
002800     . 
002900 01 WS-Datain-Post. 
003000    05 WS-In-Drycknr Pic 9(05). 
003100    05               Pic X(06). 
003200    05 WS-In-Volym   Pic 9(04). 
003300    05               Pic X(05). 
003400    05 WS-In-Pris    Pic 9(04). 
003500    05               Pic X(05). 
003600    05 WS-In-Prdatum Pic X(10). 
003700 
003800 01 Villkor          Pic 9(01) Value 0. 
003900    88 Klar                    Value 1. 
004000 
004100 01 SQL-Data. 
004200    05 WS-SQLCODE    Pic +(09). 
004300 
004400 01 Raknare. 
004500    05 Antal-Rader      Pic 9(04) Value 0. 
004600    05 Antal-Dublett    Pic 9(04) Value 0. 
004700    05 Antal-Insert     Pic 9(04) Value 0. 
004800    05 Antal-FelDrycknr Pic 9(04) Value 0. 
004900    05 Antal-AnnatFel   Pic 9(04) Value 0. 
005000 
005100 
005200 Procedure Division. 
005300     Display '-->> Programmet startar...' 
005400     Perform Oppna-Datain 
005500     Perform Laes-Datain Until Klar 
005600     Perform Stang-Datain 
005700     Perform Skriv-Inforader 
005800     Display '-->> Programmet avslutar...' 
005900 
006000     GoBack 
006100     . 
006200*--------------------------------------- 
006300 Oppna-Datain. 
006400*--------------------------------------- 
006500     Open Input DATAIN 
006600     Display '-->> Datain öppnad...' 
006700     . 
006800*--------------------------------------- 
006900 Laes-Datain. 
007000*--------------------------------------- 
007100     Read DATAIN Into WS-Datain-Post 
007200        At End 
007300           Set Klar to True 
007400        Not At End 
007500           Perform Behandla-Datain-Post 
007600     End-Read 
007700     . 

007800*--------------------------------------- 
007900 Behandla-Datain-Post. 
008000*--------------------------------------- 
008100     Add 1 to Antal-Rader 
008200     Move WS-In-Drycknr to Drycknr in Dcldryck 
008300 
008400     Exec SQL 
008500       Select 
008600          Drycknamn 
008700       Into 
008800          :Drycknamn 
008900       From 
009000          UTB00.DRYCK 
009100       Where 
009200          Drycknr = :Dcldryck.Drycknr 
009300     End-Exec 
009400 
009500     Evaluate SQLCODE 
009600        When 0 
009700           Perform SQL-Insert-Pris 
009800        When 100 
009900           Display '*** Fel Drycknr: '  
009910                    WS-In-Drycknr 
010000           Add 1 to Antal-FelDrycknr 
010100        When Other 
010200           Perform SQL-Fel 
010300     End-Evaluate 
010400     . 
010500*--------------------------------------- 
010600 SQL-Insert-Pris. 
010700*--------------------------------------- 
010800 
010900     Move Ws-In-Volym to Volym in DclPris 
011000     Move Ws-In-Pris  to Pris  in DclPris 
011100     Move Ws-In-Prdatum to Prdatum in DclPris 
011200     Exec SQL 
011300     Insert 
011400        Into UTB00.Pris 
011500     Values 
011600        (:Dcldryck.Drycknr, 
011700         :Volym, 
011800         :Pris, 
011900          Date(:Prdatum)) 
012000     End-Exec 
012100 
012200     Evaluate SQLCODE 
012300        When 0 
012400           Add 1 to Antal-Insert 
012500        When 100 
012600           Display '*** Fel Drycknr: '  
012610                    WS-In-Drycknr 
012700        When -803 
012800           Display '*** Dublett :  ' WS-Datain-Post 
012900           Add 1 to Antal-Dublett 
013000        When Other 
013100           Add 1 to Antal-AnnatFel 
013200           Perform SQL-Fel 
013300     End-Evaluate 
013400     . 
013500*--------------------------------------- 
013600 Stang-Datain. 

013700*--------------------------------------- 
013800     Close Datain 
013900     Display '-->> Datain stängd...' 
014000     . 
014100*--------------------------------------- 
014200 Skriv-Inforader. 
014300*--------------------------------------- 
014400     Display '------------------------------------' 
014500     Display 'A N T A L    Behandlade rader  : ' 
014600              Antal-Rader 
014700     Display '             Korrekta Insert   : ' 
014800              Antal-Insert 
014900     Display '             Dubletter         : ' 
015000              Antal-Dublett 
015100     Display '             Felaktiga Drycknr : ' 
015200              Antal-FelDrycknr 
015300     Display '             Andra fel         : ' 
015400              Antal-AnnatFel 
015500     Display '------------------------------------' 
015600     . 
015700*--------------------------------------- 
015800 SQL-Fel. 
015900*--------------------------------------- 
016000     Move SQLCODE to WS-SQLCODE 
016100     Display '*** SQL-Fel ***' 
016200     Display 'SQLCODE  : ' WS-SQLCODE 
016300     Display 'SQLSTATE : ' SQLSTATE 
016400     Display 'SQLMedd  : ' SQLERRMC(1:SQLERRML) 
016500     Move 99 to Return-Code 
016600     Set KLAR to True 
016700     . 
016800*--------------------------------------- 
016900 End Program OVN43. 
017000*--------------------------------------- 

Övning 5 – COBOL DELETE 
Uppgift 1 – DELETE med felhantering  

000100 Identification Division. 
000200 Program-Id. OVN51. 
000300 Environment Division. 
000400 Input-Output Section. 
000500 File-Control. 
000600     Select DATAIN Assign to DATADD 
000700         Organization is Sequential 
000800         Access is Sequential 
000900         . 
001000 Data Division. 
001100 
001200 File Section. 
001300 FD Datain Recording Mode F. 
001400 01 Datain-Post Pic X(80). 
001500 
001600 Working-Storage Section. 
001700     Exec SQL 
001800        Include SQLCA 
001900     End-Exec 
002000     . 
002100     Exec SQL 
002200        Include DRYCK 
002300     End-Exec 
002400     . 
002500     Exec SQL 
002600        Include PRIS 
002700     End-Exec 
002800     . 
002900 01 WS-Datain-Post. 
003000    05 WS-In-Drycknr Pic 9(05). 
003100    05               Pic X(06). 
003200    05 WS-In-Volym   Pic 9(04). 
003300    05               Pic X(05). 
003400    05 WS-In-Pris    Pic 9(04). 
003500    05               Pic X(05). 
003600    05 WS-In-Prdatum Pic X(10). 
003700 
003800 01 Villkor          Pic 9(01) Value 0. 
003900    88 Klar                    Value 1. 
004000 
004100 01 SQL-Data. 
004200    05 WS-SQLCODE          Pic +(09). 
004300 
004400 01 Raknare. 
004500    05 Antal-Rader      Pic 9(04) Value 0. 
004600    05 Antal-DeleteFel  Pic 9(04) Value 0. 
004700    05 Antal-Delete     Pic 9(04) Value 0. 
004800    05 Antal-FelDrycknr Pic 9(04) Value 0. 
004900    05 Antal-AnnatFel   Pic 9(04) Value 0. 
005000 
005100 
005200 Procedure Division. 
005300     Display '-->> Programmet startar...' 
005400     Perform Oppna-Datain 
005500     Perform Laes-Datain Until Klar 

005600     Perform Stang-Datain 
006300     Display '-->> Programmet avslutar...' 
006400     Perform Skriv-Summering 
006500 
006600     GoBack 
006700     . 
006800*--------------------------------------- 
006900 Oppna-Datain. 
007000*--------------------------------------- 
007100     Open Input DATAIN 
007200     Display '-->> Datain öppnad...' 
007300     . 
007400*--------------------------------------- 
007500 Laes-Datain. 
007600*--------------------------------------- 
007700     Read DATAIN Into WS-Datain-Post 
007800        At End 
007900           Set Klar to True 
008000        Not At End 
008100           Perform Behandla-Datain-Post 
008200     End-Read 
008300     . 
008400*--------------------------------------- 
008500 Behandla-Datain-Post. 
008600*--------------------------------------- 
008700     Add 1 to Antal-Rader 
008800     Move WS-In-Drycknr to Drycknr in Dcldryck 
008900 
009000     Exec SQL 
009100       Select 
009200          Drycknamn 
009300       Into 
009400          :Drycknamn 
009500       From 
009600          UTB00.DRYCK 
009700       Where 
009800          Drycknr = :Dcldryck.Drycknr 
009900     End-Exec 
010000 
010100     Evaluate SQLCODE 
010200        When 0 
010300           Perform SQL-Delete-Pris 
010400        When 100 
010500           Display '*** Fel Drycknr: '  
010510                    WS-In-Drycknr 
010600           Add 1 to Antal-FelDrycknr 
010700        When Other 
010800           Add 1 to Antal-AnnatFel 
010900           Perform SQL-Fel 
011000     End-Evaluate 
011100     . 
011200*--------------------------------------- 
011300 SQL-Delete-Pris. 
011400*--------------------------------------- 
011500 
011600     Move Ws-In-Drycknr to Drycknr in DclPris 
011700     Move Ws-In-Prdatum to Prdatum in DclPris 
011800     Move Ws-In-Volym   to Volym   in DclPris 
011900     Exec SQL 
012000     Delete 
012100        From UTB00.Pris 

012200     Where 
012300         Drycknr = :Dclpris.Drycknr 
012400         And 
012500         Prdatum = Date(:Dclpris.Prdatum) 
012600         And 
012700         Volym   = :Dclpris.Volym 
012800     End-Exec 
012900 
013000     Add 1 to Antal-Rader 
013100     Evaluate SQLCODE 
013200        When 0 
013300           Add 1 to Antal-Delete 
013400        When 100 
013500           Display '*** Fel vid DELETE ' 
013600                    WS-Datain-Post 
013700           Add 1 to Antal-DeleteFel 
013800        When Other 
013900           Display '*** Annat fel      ' 
014000                    WS-Datain-Post 
014100           Add 1 to Antal-AnnatFel 
014200           Perform SQL-Fel 
014300     End-Evaluate 
014400     . 
014500*--------------------------------------- 
014600 Stang-Datain. 
014700*--------------------------------------- 
014800     Close Datain 
014900     Display '-->> Datain stängd...' 
015000     . 
015100*--------------------------------------- 
015200 Skriv-Summering. 
015300*--------------------------------------- 
015400     Display '------------------------------------' 
015500     Display 'A N T A L    Behandlade rader  : ' 
015600              Antal-Rader 
015700     Display '             Korrekta Delete   : ' 
015800              Antal-Delete 
015900     Display '             Felaktiga Delete  : ' 
016000              Antal-DeleteFel 
016100     Display '             Felaktiga Drycknr : ' 
016200              Antal-FelDrycknr 
016300     Display '             Andra fel         : ' 
016400              Antal-AnnatFel 
016500     Display '------------------------------------' 
016600     . 
016700*--------------------------------------- 
016800 SQL-Fel. 
016900*--------------------------------------- 
017000     Move SQLCODE to WS-SQLCODE 
017100     Display '*** SQL-Fel ***' 
017200     Display 'SQLCODE  : ' WS-SQLCODE 
017300     Display 'SQLSTATE : ' SQLSTATE 
017400     Display 'SQLMedd  : ' SQLERRMC(1:SQLERRML) 
017500     Move 99 to Return-Code 
017600*    Set KLAR to True 
017700     . 
017800*--------------------------------------- 
017900 End Program OVN51. 
018000*--------------------------------------- 

